

	
	u "data/panel_mafiaconsmig", clear
	
		drop if iddecade==1|iddecade==4 // 1951 | 1981
	
**-----------------------------------------------------

			* Appendix F 
			*Robustness to main analyses in appendix

**-----------------------------------------------------
	

*Table F.1: Robust to definition of mafia
	
		su newslasso
		local meandv=r(mean)
		ivreghdfe newslasso (zempcp zmigshare zem = zssc zssmig zss) ipop, abs(municipalities iddecade) first cluster(municipalities)
		outreg2 using "tables/tableF1", replace tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(Mean DV, `meandv') nocons nor2 ctitle("News", "selected", "by lasso")

		su newsrforest
		local meandv=r(mean)
		ivreghdfe newsrforest (zempcp zmigshare zem = zssc zssmig zss) ipop, abs(municipalities iddecade) first cluster(municipalities)
		outreg2 using "tables/tableF1", tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(Mean DV, `meandv') nocons nor2 ctitle("News", "selected by", "random forest")

		su newsestseq
		local meandv=r(mean)
		ivreghdfe newsestseq (zempcp zmigshare zem = zssc zssmig zss) ipop, abs(municipalities iddecade) first cluster(municipalities)
		outreg2 using "tables/tableF1", tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(Mean DV, `meandv') nocons nor2 ctitle("News", "mafia-crimes", "only")
		
		su newspc_c
		local meandv=r(mean)
		ivreghdfe newspc_c (zempcp zmigshare zem = zssc zssmig zss) ipop,abs(municipalities iddecade) first cluster(municipalities)
		outreg2 using "tables/tableF1", tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(Mean DV, `meandv') nocons nor2 ctitle("Assuming", "continuous", "presence")
		
		su news_lag
		local meandv=r(mean)
		ivreghdfe news_lag  (zempcp zmigshare zem = zssc zssmig zss) ipop, abs(municipalities iddecade) first cluster(municipalities)
		outreg2 using "tables/tableF1", tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(Mean DV, `meandv') nocons nor2 ctitle("Mafia", "news", "t-1")
		
		su news_winsor
		local meandv=r(mean)
		ivreghdfe news_winsor (zempcp zmigshare zem = zssc zssmig zss) ipop, abs(municipalities iddecade) first cluster(municipalities)
		outreg2 using "tables/tableF1", tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(Mean DV, `meandv') nocons nor2 ctitle("News", "winsor-", "ized")
		
		gen lnewspc=ln(newspc+1)
		su lnewspc
		local meandv=round(r(mean), 0.001)
	ivreghdfe lnewspc (zempcp zmigshare zem = zssc zssmig zss) ipop, abs(municipalities iddecade) first cluster(municipalities)	 
	outreg2 using "tables/tableF1", tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(Mean DV, `meandv') nocons nor2 ctitle("Log", "Mafia", "news")


*Table F.2: Robustness using variations in the instruments and specification

	* Drought severity
	
	 ivreghdfe newspc (zempcp zmigshare zem = zssc zssmig_rob zss_rob) ipop, abs(municipalities iddecade) first cluster(municipalities)
		mat t=e(first)
		local fs1=round(t[4,1], 0.01)
		local fs2=round(t[4,2], 0.01)
		local fs3=round(t[4,3], 0.01)
		local f1=round(t[8,1], 0.01)
		local f2=round(t[8,2], 0.01)
		local f3=round(t[8,3], 0.01)
		local ar=round(e(archi2), 0.1)
	outreg2 using "tables/tableF2", replace tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) adds(SW F-Stat, `f3', A-R Wald test, `ar') nocons nor2 ctitle("Using Drought", "as push for Migr")
	
	* RD Bandwidth
	
	rdrobust zempcp running, vce(cluster municipalities)
		local bdw=e(h_l)
		
	ivreghdfe newspc (zempcp zmigshare zem = zssc zssmig zss) if pop1871>=10000-`bdw' & pop1871<=10000+`bdw', abs(municipalities iddecade) first savefirst cluster(municipalities) 
		mat t=e(first)
		local f3=round(t[8,3], 0.01)
		local ar=round(e(archi2), 0.1)
		local ncities=e(N_clust)
		outreg2 using "tables/tableF2", tex label dec(3) keep(zempcp zmigshare zem) addtext(Number of municipalities, `ncities', City Decade FE, Yes, SW F-Stat, `f3', A-R Wald test, `ar') nocons nor2 ctitle("RDD","bandwidth")

	*Construction without leave out (wlo)
	ivreghdfe newspc ipop (zempcp zmigshare zem = zssc_wlo zssmig zss_wloc), abs(municipalities iddecade) first savefirst cluster(municipalities)
	mat t=e(first)
	local f3=round(t[8,3], 0.01)
	local ar=round(e(archi2), 0.1)
	local ncities=e(N_clust)
	outreg2 using "tables/tableF2", tex label dec(3) keep(zempcp zmigshare zem) addtext(Number of municipalities, `ncities', City Decade FE, Yes) adds(SW F-Stat, `f3', A-R Wald test, `ar') nocons nor2  ctitle("IV Constr w/o", "leave out")
	
	*Shift-share without leave out (wlo)
	ivreghdfe newspc ipop (zempcp zmigshare zem = zssc_wlo zssmig_wlo zss_wlo2), abs(municipalities iddecade) first savefirst cluster(municipalities)
	mat t=e(first)
	local f3=round(t[8,3], 0.01)
	local ar=round(e(archi2), 0.1)
	local ncities=e(N_clust)
	outreg2 using "tables/tableF2", tex label dec(3) keep(zempcp zmigshare zem) addtext(Number of municipalities, `ncities', City Decade FE, Yes, SW F-Stat, `f3', A-R Wald test, `ar') nocons nor2 ctitle("IV Constr \& Migr", " w/o leave out")
		
	*Migshare_currentpop
	drop zmigshare zem
	rename (zmigshare_currentpop zem_currentpop)(zmigshare zem)
	ivreghdfe newspc (zempcp zmigshare zem =  zssc zssmig zss) ipop, abs(municipalities iddecade) first savefirst cluster(municipalities)
	mat t=e(first)
	local f3=round(t[8,3], 0.01)
	local ar=round(e(archi2), 0.1)
	local ncities=e(N_clust)
	outreg2 using "tables/tableF2", tex label dec(3) keep(zempcp zmigshare zem) addtext(Number of municipalities, `ncities', City Decade FE, Yes,  SW F-Stat, `f3', A-R Wald test, `ar') nocons nor2 ctitle("Endogenous", "population")
	
	*including Molise and Abruzzo in south
	preserve 
	drop zmigshare zem
	rename (zmigshare_abruzzo zem_abruzzo)(zmigshare zem)
	 ivreghdfe newspc (zempcp zmigshare zem = zssc zssmig zss) ipop, abs(municipalities iddecade) first savefirst cluster(municipalities) 
	 mat t=e(first)
	local f3=round(t[8,3], 0.01)
	local ar=round(e(archi2), 0.1)
	local ncities=e(N_clust)
	 outreg2 using "tables/tableF2", tex label dec(3) keep(zempcp zmigshare zem) addtext(Number of municipalities, `ncities', City Decade FE, Yes,  SW F-Stat, `f3', A-R Wald test, `ar') nocons nor2 ctitle("Consider Abruzzo", "Molise as south")
	
	restore	
	

*Table F3: Table F.3: Robustness using alternative methods to compute standard errors

		
	*Spatial HAC 200 km
	acreg newspc dec2-dec3 ipop  (zempcp zmigshare zem = zssc zssmig zss), id(municipalities) time(iddecade) latitude (latitude) longitude(longitude) dist(200) lag(2) spatial hac pfe1(municipalities) pfe2(iddecade) 
outreg2 using "tables/tableF3.tex", replace dec(3) label nocons nor2 keep(zempcp zmigshare zem) ctitle("Spatial HAC", "200 km")  addtext(City Decade FE, Yes) 

	*Spatial HAC 300 km
	acreg newspc dec2-dec3 ipop  (zempcp zmigshare zem = zssc zssmig zss), id(municipalities) time(iddecade) latitude (latitude) longitude(longitude) dist(300) lag(2) spatial hac pfe1(municipalities) pfe2(iddecade) 
outreg2 using "tables/tableF3.tex", dec(3) label nocons nor2 keep(zempcp zmigshare zem) ctitle("Spatial HAC", "300 km")  addtext(City Decade FE, Yes) 
	
*Bootstrapping 
	xtset municipalities iddecade
	 tsset, clear
	bootstrap, r(100) cl(municipalities) id(gnew2) seed(12345): xtivreg2 newspc dec2-dec3 (zempcp zmigshare zem = zssc zssmig zss) ipop, first cl(municipalities) fe i(municipalities) 
	outreg2 using "tables/tableF3.tex", dec(3) label nocons nor2 keep(zempcp zmigshare zem) ctitle("Bootstrap")  addtext(City Decade FE, Yes)
	
*Cluster at the province level
	ivreghdfe newspc ipop (zempcp zmigshare zem = zssc zssmig zss), abs(municipalities iddecade) first savefirst cluster(cod_prov)
	mat t=e(first)
	local f3=round(t[8,3], 0.01)
	local ar=round(e(archi2), 0.1)
	outreg2 using "tables/tableF3.tex", tex label dec(3) keep(zempcp zmigshare zem) addtext(City Decade FE, Yes) nocons nor2 ctitle("Cluster province", "level")

	

*Figure F.1: Validation of La Stampa as a comparable source of news to Corriere

u "data/validation_corriere_1969.dta", clear

replace n_corriere=0 if n_corriere==.
replace n_lastampa=0 if n_lastampa==.

ivreg2 n_corriere (zempcp zmigshare zem = zssc zssmig zss), cluster(PRO_COM) first
		local fstat=round(e(first)[8,3],0.001)
		local N=e(N)
		local b1=round(e(b)[1,3],0.001)
eststo corriere

ivreg2 n_lastampa (zempcp zmigshare zem = zssc zssmig zss), cluster(PRO_COM) first
		local b2=round(e(b)[1,3],0.001)
eststo lastampa


coefplot (corriere, color(black)) (lastampa, ciopts(color(gray)) color(gray)), drop(_cons zempcp zmigshare) legend(off) citop vertical xscale(ext) byopts(note("A")) text(7 0.85 "Corriere" "beta=`b1'" "N=`N'" "F=`fstat'", justification(left) placement(e) size(small)) text(7 1.2 "La Stampa" "beta=`b2'" "N=`N'" "F=`fstat'", justification(left) placement(e) size(small)) ytitle("Effect on mafia news in 1969") yline(0, lpattern(dash) lcolor(red))
graph export "figures/figureF1.png", replace
 
 
